Komfortable Druckersteuerung per VBA Teil 1
Versionen: Access 2003 und 2002/XP

Ab Access 2002/XP stehen Ihnen erweiterte Möglichkeiten für den Umgang mit Druckern zur Verfügung: Das Printer-Objekt und die Printers-Auflistung sowie eine ganze Reihe spezieller Methoden und Eigenschaften für die Druckausgabe von Formularen und Berichten. In diesem Tipp der Woche zeigen wir Ihnen, wie Sie diese neuen Möglichkeiten effektiv nutzen und zum Beispiel Berichte gezielt auf individuell wählbare Drucker ausgeben oder das Problem mit willkürlich verstellten Druckrändern lösen.

Beginnen wir mit einigen Grundlagen: Basis aller erweiterten Möglichkeiten ist das neue Printer-Objekt. Es zahlreiche Eigenschaften, über die sich aktuelle Einstellungen eines Druckers abrufen oder individuell einstellen lassen. Im Großen und Ganzen spiegelt das Printer-Objekt die Einstellungen wieder, die Sie über das Menü Datei-Seite einrichten vornehmen können. Zusätzlich dazu stehen erweiterte Eigenschaften für die Ansteuerung spezieller Features wie beispielsweise Duplex-Druck, Anzahl zu druckender Kopien, Auswahl des Papierschachtes oder zur Umschaltung des Farb-/Qualitäts-Modus zur Verfügung.

Die Basiseinstellungen aller installierten Drucker stellt die Printers-Auflistung zur Verfügung. Sie enthält pro installiertem Drucker ein Printer-Objekt mit den Einstellungen, die für den betreffenden Drucker über Systemsteuerung/Drucker und über Datei-Seite einrichten als Standard festgelegt wurden.

Analog dazu verfügen Formulare und Berichte über eine Printer-Eigenschaft, über die Sie den zu verwendenden Drucker und dessen Einstellungen spezifizieren können. Schließlich stellt Access selbst über das Application-Objekt eine Printer-Eigenschaft zur Verfügung, die dem jeweils eingestellten Standard-Drucker entspricht und über die Sie den als Standard zu verwendenden Drucker und dessen Einstellungen sowohl auslesen als auch festlegen können.

Anhand verschiedener kleiner Beispiele möchten wir den Einsatz der Möglichkeiten des Printer-Objektes und der Printers-Auflistung demonstrieren.

Beispiel 1: Das Formular "Standard-Drucker wechseln" soll eine Auswahl installierter Drucker in einem Kombinationsfeld "clDrucker" zur Verfügung stellen, über das einer der Drucker als neuer Standard-Drucker gesetzt werden kann. Per Mausklick auf eine Schaltfläche Setzen (btnSetStd) wird der im Kombinationsfeld eingestellte Drucker als Access-Standard-Drucker gesetzt. Die Einstellungen über Systemsteuerung/Drucker bezüglich des von Windows zu verwendenden Standard-Druckers bleiben davon unberührt! Der hier gewählte Standard-Drucker kommt nur innerhalb von Access für alle Druckausgaben zum Einsatz und für Berichte, in denen über das Menü Datei-Seite einrichten auf der Registerkarte "Seite" die Option "Standarddrucker" eingestellt wurde.

Das Formular "Standard-Drucker wechseln" bauen Sie wie folgt auf:

· Legen Sie ein neues Formular an.

· Platzieren Sie zwei Bezeichnungsfelder am oberen Rand. Das erste erhält die Beschriftung "Aktueller Standarddrucker", das zweite nennen Sie "txtStandard".

· Legen Sie darunter ein Bezeichnungsfeld und ein Kombinationsfeld an. Das Bezeichnungsfeld erhält die Beschriftung "Neuer Standard-Drucker", das Kombinationsfeld nennen Sie "clDrucker".

· Unter dem Kombinationsfeld legen Sie nun eine Schaltfläche mit der Beschriftung Setzen und dem Namen "btnSetStd" an und richten es so aus, dass der rechte Rand der Schaltfläche mit dem linken Rand des Kombinationsfeldes bündig ist.

· Stellen Sie die Eigenschaft "Beim Laden" des Formulars auf "[Ereignisprozedur]" ein, klicken Sie auf die Schaltfläche mit den drei Punkten und geben Sie folgende Anweisungen im VBA-Editor ein:

Private Sub Form_Load()
  Dim prtLoop As Printer

  Me.txtStandard = Application.Printer.DeviceName

  With Me.clDrucker
    For Each prtLoop In Application.Printers
      .AddItem prtLoop.DeviceName
    Next 'prtLoop
  End With
  Me.clDrucker = Me.txtStandard

End Sub

Der jeweils gesetzte Standard-Drucker lässt sich über die Eigenschaft "Application.Printer.DeviceName" auslesen. "Application.Printer" zeigt immer auf das Printer-Objekt, das dem aktuellen Standard-Drucker entspricht. Die Eigenschaft "DeviceName" liefert dann den betreffenden Namen, so wie er beispielsweise auch im Dialog Datei-Drucken in der Liste Name angezeigt wird. Um die installierten Drucker im Kombinationsfeld anzuzeigen, lesen wir die Printers-Auflistung beim Öffnen des Formulars aus und fügen die einzelnen Druckernamen der Liste über deren Methode "AddItem" hinzu. Dazu deklarieren wir eine Objektvariable "prtLoop" vom Datentyp "Printer", die anschließend in einer "For..Each"-Schleife für den Zugriff auf die einzelnen Printer-Objekte in der Printers-Auflistung verwendet wird. Für die Identifikation eines Druckers ist immer dessen Name aus "DeviceName" ausschlaggebend, so dass wir diesen Namen hier zur Liste hinzufügen. Sind alle Drucker ausgelesen, setzen wir die Auswahl in der Liste abschließend auf den aktuellen Standard-Drucker.

Alternativ zur "For..Each"-Schleife können Sie auch eine "normale" Schleife mit Zählvariable einsetzen. Für die Terminierung der Schleife kann die Eigenschaft "Printers.Count" ausgelesen werden, die Auskunft über die Anzahl installierter Drucker gibt:

Sub Test1()
  Dim intAnz As Integer, I As Integer

  intAnz = Printers.Count - 1
  For I = 0 To intAnz
    Debug.Print Printers(I).DeviceName
  Next I

End Sub

Die Printers-Auflistung wird, wie andere Auflistungen auch, über eine Indexnummer beginnend mit "0" adressiert, so dass wir hier mit "0" beginnen und "Count-1" über "intAnz" zugrunde legen müssen.

· Stellen Sie die Eigenschaft "Beim Klicken" der Schaltfläche auf "[Ereignisprozedur]" ein, klicken Sie auf die Schaltfläche mit den drei Punkten und geben Sie folgende Anweisungen im VBA-Editor ein:

Private Sub btnSetStd_Click()
  Dim strPrinter As String
  Dim prtNew As Printer

  On Error Resume Next
  strPrinter = Me.clDrucker
  If Err <> 0 Then
    Beep
    Exit Sub
  End If
  On Error GoTo 0

  'Gewählten Drucker auslesen...
  Set prtNew = Printers(strPrinter)
   '...und als Standard setzen
  Set Application.Printer = prtNew

  Me.txtStandard = Application.Printer.DeviceName

End Sub

Wie erwähnt, beinhaltet "Application.Printer" ein Printer-Objekt, das den aktuell eingestellten Standard-Drucker repräsentiert. Um einen neuen Standard-Drucker zu setzen, muss das aktuelle Printer-Objekt ersetzt werden. Dazu müssen wir ein neues, temporäres Printer-Objekt, hier deklariert als "prtNew" einsetzen. In dieses Printer-Objekt lesen wir das zum ausgewählten Drucker gehörende Printer-Objekt über "Set prtNew = Printers(strPrinter)" aus der Printers-Auflistung aus. Die Zuweisung "Set Application.Printer = prtNew" sorgt schließlich dafür, dass Access von dem neuen Standard-Drucker Kenntnis erhält.

· Speichern Sie das Formular unter dem Namen "Standard-Drucker wechseln".

Dieses Formular können Sie nun beispielsweise aus einem "Menüformular" für den Ausdruck Ihrer Berichte über eine Schaltfläche Standarddrucker Ändern (btnStdPrt) aufrufen und Ihren Anwendern so die Umschaltung vereinfachen:

Private Sub btnStdPrt_Click()

  DoCmd.OpenForm "Standard-Drucker wechseln"

End Sub

Im zweiten Teil dieses Tipps der Woche zeigen wir, wie Sie Berichte auf individuell gewählte Drucker ausgeben können.

Im ersten Teil dieses Tipps der Woche hatten wir zur Demonstration der Möglichkeiten der Printer-Objekte eine Lösung zur einfacheren Umschaltung des Standarddruckers vorgestellt. In einem 2. Beispiel möchten wir uns nun einmal ansehen, wie Berichte auf einen bestimmten Drucker ausgedruckt werden können.

Wie im ersten Teil bereits kurz angerissen, verfügen Berichte (und Formulare) nun über eine Eigenschaft "Printer", die das Printer-Objekt des für den Ausdruck zu verwendenden Druckers repräsentiert. Wurde für den Bericht über Datei-Seite Einrichten auf der Registerkarte "Seite" die Option "Standarddrucker" eingestellt, setzt Access das Printer-Objekt des Berichtes auf "Application.Printer". Abfragen können Sie diese Einstellung übrigens per VBA über die Eigenschaft "UseDefaultPrinter" des Berichtes. Wurde für den Bericht über Datei-Seite Einrichten auf der Registerkarte "Seite" die Option Spezieller Drucker eingestellt und ein entsprechender Drucker gewählt, speichert Access den Bericht mit dem zum ausgewählten Drucker gehörenden, aus der Printers-Auflistung ausgelesen Printer-Objekt.

Um nun einen Bericht auf einem anderen, als den über Datei-Seite Einrichten eingestellten Drucker - egal, ob Standard oder Spezial - auszudrucken, muss das Printer-Objekt des Berichtes vor dem Ausdruck durch das Printer-Objekt des zu verwendenden Druckers ersetzt werden. Diese Aufgabe soll ein Formular "Bericht/Drucker-Auswahl" übernehmen:

· Legen Sie ein neues Formular an.

· Platzieren Sie ein Listenfeld "lstDrucker" im Formular, geben Sie für das Bezeichnungsfeld "Bericht:" ein und setzen Sie die folgenden Eigenschaften für das Listenfeld:

Herkunftstyp: Werteliste
Anzahl Spalten: 2
· Stellen Sie die Eigenschaft "Beim Klicken" der Liste auf "[Ereignisprozedur]" ein und klicken Sie auf die Schaltfläche mit den drei Punkten, um in die VBA-Entwicklungsumgebung zu wechseln.

· Wählen Sie Extras-Verweise an und setzen Sie eine Referenz auf "Microsoft DAO Object Library 3.6".

· Kopieren Sie folgende Anweisungen in die Ereignisprozedur:

  Dim strPrinter As String

  On Error Resume Next
  strPrinter = Me.lstBerichte.Column(1)
  If Err <> 0 Then
    Beep
    Exit Sub
  End If
  On Error GoTo 0

  If strPrinter = "Standard" Then
    strPrinter = Me.txtStandard
  End If
  Me.clDrucker = strPrinter

· Legen Sie ein Kombinationsfeld "clDrucker" an, geben Sie für dessen Bezeichnungsfeld "Drucker:" ein und stellen Sie die Eigenschaft "Herkunftstyp" auf "Werteliste".

· Legen Sie ein Textfeld "txtStandard" an und geben Sie für das Bezeichnungsfeld "Standard:" ein.

· Platzieren Sie eine Schaltfläche "btnDruck" im Formular und setzen Sie die Eigenschaft "Beschriftung" auf Drucken.

· Stellen Sie die Eigenschaft "Beim Klicken" der Schaltfläche auf "[Ereignisprozedur]", klicken Sie auf die Schaltfläche mit den drei Punkten und kopieren Sie folgende Anweisungen in die Ereignisprozedur:

  Dim R As Report, strReport As String,
  Dim strPrinter As String, strStdPrinter As String
  Dim prtCurr As Printer

  On Error Resume Next
  strReport = Me.lstBerichte.Value
  If Err <> 0 Then
    Beep
    Exit Sub
  End If
  Err = 0
  strPrinter = Me.clDrucker
  If Err <> 0 Then
    Beep
    Exit Sub
  End If
  On Error GoTo 0

  strStdPrinter = Application.Printer.DeviceName
  DoCmd.OpenReport strReport, acViewPreview, , , acHidden
  Set R = Reports(strReport)
  If R.UseDefaultPrinter And _
     strPrinter = strStdPrinter Then GoTo PrtOK

  'Gewählten Drucker auslesen...
  Set prtCurr = Printers(strPrinter)
  '...und für Bericht setzen
  Set R.Printer = prtCurr

PrtOK:
  If Me.cbPreview Then
    DoCmd.OpenReport strReport, acViewPreview
  Else
    On Error Resume Next
    Application.Echo False
    DoCmd.SelectObject acReport, strReport, False
    DoCmd.PrintOut
    DoCmd.Close acReport, strReport
    Application.Echo True
  End If

· Legen Sie unter der Schaltfläche ein Kontrollkästchen "cbPreview" an und geben Sie "Vorschau" für die Beschriftung ein.

· Platzieren Sie eine Schaltfläche "btnDone" im Formularfuß und geben Sie als Beschriftung Fertig ein.

· Stellen Sie die Eigenschaft "Beim Klicken" der Schaltfläche auf "[Ereignisprozedur]", klicken Sie auf die Schaltfläche mit den drei Punkten und kopieren Sie folgende Anweisungen in die Ereignisprozedur:

On Error Resume Next
DoCmd.Close acForm, Me.Name, acSavePrompt
· Stellen Sie die Eigenschaft "Beim Laden" des Formulars auf "[Ereignisprozedur]", klicken Sie auf die Schaltfläche mit den drei Punkten und kopieren Sie folgende Anweisungen in die Ereignisprozedur:

  Dim db As DAO.Database
  Dim conTmp As DAO.Container
  Dim docTmp As DAO.Document
  Dim strX As String, strReport As String, R As Report
  Dim prtLoop As Printer

  Me.txtStandard = Application.Printer.DeviceName

  Set db = CurrentDb()
  Set conTmp = db.Containers("Reports")
  With Me.lstBerichte
    For Each docTmp In conTmp.Documents
      strReport = docTmp.Name
      DoCmd.OpenReport strReport, _
                       acViewPreview, , , acHidden
      Set R = Reports(strReport)
      If R.UseDefaultPrinter Then
        strX = strReport & ";Standard"
      Else
        strX = strReport & ";" & R.Printer.DeviceName
      End If
      DoCmd.Close acReport, strReport, acSaveNo
      .AddItem strX
    Next docTmp
  End With
  Me.lstBerichte = Me.lstBerichte.Column(0, 0)

  With Me.clDrucker
    For Each prtLoop In Application.Printers
      .AddItem prtLoop.DeviceName
    Next 'prtLoop
  End With
  Me.clDrucker = Me.txtStandard

· Speichern Sie das Formular unter dem Namen "Bericht/Drucker-Auswahl" und lassen Sie es testweise anzeigen.

Zur Funktionswiese der Lösung: Das Formular zeigt zwei Listen mit verfügbaren Berichten und installierten Druckern an. Nach Auswahl des Berichtes und des gewünschten Druckers startet ein Klick auf die Schaltfläche Drucken die Ausgabe auf den betreffenden Drucker oder - wenn die Option Vorschau aktiviert ist - die Anzeige in der Seitenansicht auf dem Bildschirm. Die Ereignisprozedur "Beim Klicken" der Schaltfläche Drucken arbeitet wie folgt: Zunächst lesen wir den gewählten Bericht und den dafür eingestellten Drucker aus. Kommt es dabei zu einem Fehler, weil eins von beiden noch nicht ausgewählt wurde, so geben wir einen akustischen Hinweis aus und verlassen die Routine. Für den späteren Vergleich lesen wir dann den aktuell gesetzten Standard-Drucker aus.

Wie bereits angedeutet, muss das Printer-Objekt des Berichtes durch das Printer-Objekt des zu verwendenden Drucker ersetzt werden. Dazu öffnen wir den Bericht in der Vorschau und im Modus "Versteckt", um ein störendes Flackern des Bildschirms zu vermeiden. Dann setzen wir eine Referenz auf den Bericht in der Objektvariablen "R", um die folgenden Zugriffe zu vereinfachen. Im Anschluss prüfen wir, ob die Ausgabe auf den Standard-Drucker erfolgen soll und ob dieser als Ausgabe-Drucker gewählt wurde. Ist das der Fall, sind keine weiteren Aktionen notwendig, da das Printer-Objekt des Berichtes bereits korrekt gesetzt ist. Andernfalls müssen wir das Printer-Objekt des Berichtes durch das Printer-Objekt des gewählten Druckers ersetzen. Dazu verwenden wir ein neues, temporäres Printer-Objekt "prtCurr", das wir über "Set prtCurr = Printers(strPrinter)" zunächst entsprechend initialisieren. Über "Set R.Printer = prtCurr " weisen wir das Printer-Objekt dann dem Bericht zu. Je nach Option Vorschau zeigen wir dann den Bericht in der Seitenansicht an oder selektieren den versteckten Bericht, drucken ihn aus und schließen ihn dann wieder.

In der Liste "Berichte" zeigen wir neben den Namen der Berichte den jeweils eingestellten Drucker an. Diese Informationen lesen wir beim Öffnen des Formulars wie folgt aus:

  Set db = CurrentDb()
  Set conTmp = db.Containers("Reports")
  With Me.lstBerichte
    For Each docTmp In conTmp.Documents
      strReport = docTmp.Name
      DoCmd.OpenReport strReport, _
                       acViewPreview, , , acHidden
      Set R = Reports(strReport)
      If R.UseDefaultPrinter Then
        strX = strReport & ";Standard"
      Else
        strX = strReport & ";" & R.Printer.DeviceName
      End If
      DoCmd.Close acReport, strReport, acSaveNo

Die Reports-Auflistung des Database.Containers-Objektes liefert hier zunächst die Namen der einzelnen Berichte. Diese öffnen wir in der Vorschau und versteckt, setzen eine Referenz darauf und prüfen die Eigenschaft "UseDefaultPrinter". Ist diese "True", so verwenden wir "Standard" als Drucker, andernfalls lesen wir über "Printer.DeviceName" den Namen des speziellen Druckers aus.

Wie Sie sehen, ist es Dank der Printer-Objekte von Access 2003 und 2002/XP relativ einfach, einen Bericht auf einen individuell gewählten Drucker auszugeben, indem die Eigenschaft "Printer" entsprechend gesetzt wird. Ähnlich einfach können Sie weitere Eigenschaften wie beispielsweise die Seitenränder einstellen und so das Problem mit sich selbsttätig verstellenden Seitenrändern lösen - mehr dazu im dritten Teil dieses Tipps der Woche in der nächsten Ausgabe von Access Weekly.

Vor allem weniger erfahrene Anwender verstellen oftmals irrtümlich den Standard-Drucker von Access. Wird dann ein Bericht, für den "Standarddrucker" eingestellt ist, ausgedruckt, kann das recht teuer werden, wenn beispielsweise eine einfache Liste statt auf einem Tintenstrahldrucker auf einem Farblaser ausgegeben wird.

Hier lässt sich mit den Printer-Objekten Abhilfe schaffen, indem Sie beispielsweise über ein "AutoExec"-Makro den Standard-Drucker beim Start von Access zurücksetzen. Dazu nutzen Sie die Funktion "SetStdPrinter()", die als Parameter den Namen des zu setzenden Standard-Druckers erwartet:

Function SetStdPrinter(strPrinter As String)
  Dim prtNew As Printer

  On Error Resume Next
  'Gewünschten Drucker auslesen...
  Set prtNew = Printers(strPrinter)
  '...und als Standard setzen
  Set Application.Printer = prtNew

End Function

Hier wird mit dem als Parameter "strPrinter" angegebenen Drucker ein Printer-Objekt initialisiert und als Standard-Drucker gesetzt.

Um den Aufruf zu automatisieren, legen Sie ein "AutoExec"-Makro an und fügen dort die Aktion "AusführenCode" ein oder ergänzen ein vorhandenes "AutoExec"-Makro um diese Aktion. Als Parameter "Funktionsname" geben Sie "SetStdPrinter("DruckerName")" an, wobei Sie "DruckerName" durch den Namen des zu setzenden Druckers ersetzen. Beim Start von Access wird nun automatisch "AutoExec" ausgeführt, das dann die Funktion "SetStdPrinter()" mit dem Namen des zu setzenden Druckers aufruft. Den genauen Namen bringen Sie in Erfahrung, indem Sie Datei-Drucken anwählen und in der Dropdown-Liste Name nachschlagen.

Die Funktion "SetStdPrinter()" können Sie natürlich auch von anderen Stellen aus aufrufen, um den Standard-Drucker je nach Bedarf zurückzusetzen. Wenn Sie beispielsweise ein "Menü"-Formular für den Abruf von Berichten einsetzen, könnten Sie dort in der Ereignisprozedur "Beim Laden" den Aufruf wie folgt einbauen:

Sub Form_Load()
  Dim R As Variant

  .....
  R = SetStdPrinter("HP DeskJet 710C")
  .....

End Sub

Ein weiteres Problem, dass sich bei Berichten regelmäßig zeigt, lässt sich ebenfalls per Printer-Objekt lösen: Die willkürliche Verstellung der Seitenränder. Dazu stellt das Printer-Objekt die Eigenschaften "TopMargin" (oberer Rand), "BottomMargin" (unterer Rand), "LeftMargin" (linker Rand) und "RightMargin" (rechter Rand) zur Verfügung. Diese Eigenschaften können Sie nun beim Öffnen des Berichtes über die Ereignisprozedur "Beim Öffnen" (Report_Open) beispielsweise wie folgt einstellen:

Private Sub Report_Open(Cancel As Integer)

  With Me.Printer
    .TopMargin = 850
    .BottomMargin = 1100
    .LeftMargin = 1400
    .RightMargin = 850
  End With

End Sub

Die Angaben erfolgen jeweils in der Maßeinheit "Twips". Um Sie in der etwas handlicheren Maßeinheit "Millimeter" anzugeben, kopieren Sie folgende Umrechnungsfunktion in ein neues oder in ein vorhandenes Modul:

Public Function mm(Millimeter As Single) As Long

  mm = CLng(Millimeter * 56.7)

End Function

In Zukunft können Sie dann die Ränder beim Öffnen des Berichtes beispielsweise wie folgt einstellen:

Private Sub Report_Open(Cancel As Integer)

  With Me.Printer
    .TopMargin = mm(20)
    .BottomMargin = mm(30)
    .LeftMargin = mm(25)
    .RightMargin = mm(15)
  End With

End Sub

Standarddrucker per VBA ändern
Versionen: Access 2010, 2007, 2003 und 2002/XP

 FRAGE:  Ich möchte in einem "Druck"-Formular, das verschiedene Schaltflächen mit "DoCmd.OpenReport"-Anweisungen enthält, sicherstellen, dass der Ausdruck immer auf einem bestimmten Drucker erfolgt. Ist es möglich, dazu den in Access eingestellten Standarddrucker per VBA zu ändern?
J. Dänker

 ANTWORT:  Nehmen Sie in die Ereignisprozedur "Beim Laden" (Form_Load) Ihres "Druck"-Formulars folgende Anweisungen auf:

  Dim prt As Printer

  Set prt = Application.Printers("Druckername")
  Set Application.Printer = prt

"Druckername" ersetzen Sie dabei durch den Namen des Druckers, so wie er in der Systemsteuerung oder im Dialog "Drucken" angezeigt wird. Alternativ liefert folgende Schleife in einer kleinen Testprozedur die Namen aller installierten Drucker:

Sub ListPrinterNames()
  Dim prt As Printer

  For Each prt In Printers
    Debug.Print prt.DeviceName
  Next

End Sub
15.04.2011
C:\_CD_brennen_div\Autocad\2006\Installationsanleitung AutoCad LT 2006 mit Menu mj.doc
1 von 8

